
DIRECTORY
Mopcodes: FROM "mopcodes";

DEFINITIONS FROM Mopcodes;

InlineDefs: DEFINITIONS =
BEGIN -- byte codes for inline

COPY: PROCEDURE [from: POINTER, nwords: CARDINAL, to: POINTER] =
MACHINE CODE BEGIN zBLT END;

DIVMOD: PROCEDURE [num, den: CARDINAL] RETURNS [quotient, remainder: CARDINAL] =
MACHINE CODE BEGIN zDIV; zPUSH END;

PUSH: PROCEDURE RETURNS [WORD] = MACHINE CODE BEGIN zPUSH END;

LDIVMOD: PROCEDURE [numlow: WORD, numhigh: CARDINAL, den: CARDINAL]
RETURNS [quotient, remainder: CARDINAL] =
MACHINE CODE BEGIN zLDIV; zPUSH END;

LongCARDINAL: TYPE = MACHINE DEPENDENT RECORD [
lowbits: CARDINAL,
highbits: CARDINAL];

LongMult: PROCEDURE [CARDINAL, CARDINAL] RETURNS [product: LongCARDINAL] =
MACHINE CODE BEGIN zMUL; zPUSH END;

LongDiv: PROCEDURE [num: LongCARDINAL, den: CARDINAL] RETURNS [CARDINAL] =
MACHINE CODE BEGIN zLDIV END;

LongDivMod: PROCEDURE [num: LongCARDINAL, den: CARDINAL]
RETURNS [quotient, remainder: CARDINAL] =
MACHINE CODE BEGIN zLDIV; zPUSH END;

BitOp: TYPE = PROCEDURE [WORD, WORD] RETURNS [WORD];

BITAND: BitOp = MACHINE CODE BEGIN zAND END;
BITOR: BitOp = MACHINE CODE BEGIN zOR END;
BITXOR: BitOp = MACHINE CODE BEGIN zXOR END;
BITNOT: PROCEDURE [WORD] RETURNS [WORD] = MACHINE CODE BEGIN zLIN1; zXOR END;
BITSHIFT: PROCEDURE [value: WORD, count: INTEGER] RETURNS [WORD] =
MACHINE CODE BEGIN zSHIFT END;

END...

